

# SCIENCE & VIE

ISSN 0036 8369

MENSUEL

N° 835 AVRIL 1987



## D.A.T. CONTRE COMPACT

**DÉCHETS**

**TRÈS RADIOACTIFS**

**SOUS LA FRANCE**

**TRANQUILLE**

M 2578 - 835 - 16,00 F





# La phrase cachée

INFORMATIQUE AMUSANTE

Voici un jeu de société inspiré de l'un de ceux proposés à ses débuts par la cinquième chaîne de T.V.. Il s'agit du jeu de la phrase cachée.

Il faut, un "juge" et trois concurrents. En premier lieu, sans que les participants le voient, le juge introduit dans l'ordinateur une phrase à deviner, ainsi qu'une voyelle quelconque qu'il aura choisie. Dès lors, une série de points s'affichera sur le tableau principal, indiquant ainsi le nombre de lettres à chaque mot. La voyelle choisie apparaîtra à l'emplacement qu'elle occupe respectivement dans chacun des mots. Le jeu peut commencer.

Le premier joueur lance la roue. A l'issue de ce lancer, le joueur

verra son score crédité, au hasard, d'une somme variant de 500 et 5 000 F à moins qu'il ne tombe sur la case "PATATRAC" auquel cas, quel que soit son crédit antérieur, il se retrouvera avec une somme nulle et devra passer la main. Ensuite l'ordinateur lui demandera une consonne. Celle-ci fera gagner 100 F supplémentaires chaque fois qu'elle sera présente dans la phrase et sera inscrite aux emplacements adéquats sur le tableau principal.

Enfin l'ordinateur demandera si le joueur a une idée de la phrase. Si la réponse est non ce sera au joueur suivant de lancer la roue puis de jouer. En cas contraire, la phrase sera tapée au clavier et

contrôlée. Si la réponse fournie est exacte, le joueur sera déclaré vainqueur, sinon il passera la main. Notons que, si après avoir lancé la roue, le joueur se trouve à la tête d'un crédit supérieur à 4 000 F, il lui sera possible d'acheter une voyelle; son score se trouvera alors amputé de cette somme. Ces quelques règles établies passons à l'écriture du programme.

En premier lieu, nous demandons à l'Amstrad de passer en mode 1 et la couleur du tour de l'écran sera fixée (ligne 10). Deux enveloppes sonores seront déterminées (lignes 50 à 60), puis 5 fenêtres mises en place (lignes 100 à 140). Elles correspondent respectivement à la zone d'affichage de la phrase, à celle de la roue et enfin aux trois pupitres des participants. Deux tableaux de mémorisation seront créés à la ligne 200.

Le premier, P\$ sera utilisé pour mémoriser la phrase introduite par

```

10 MODE 1:PAPER 3:BORDER 9:CLS
20 REM *****
30 REM * DETERMINATION DES ENVELOPES DU SON *
40 REM *****
50 ENV 1,1,15,1,1,0,1,15,-1,1
60 ENV 2,1,15,2,1,0,2,15,-1,2
70 REM *****
80 REM * MISE EN PLACE DES FENETRES *
90 REM *****
100 WINDOW #1,5,35,2,12
110 WINDOW #2,15,25,14,18
120 WINDOW #3,2,12,21,25
130 WINDOW #4,15,25,21,25
140 WINDOW #5,28,38,21,25
150 PAPER #1,0:PAPER #2,0:CLS #1:CLS #2
160 PAPER #3,1:PAPER #4,1:PAPER #5,1:CLS #3:CLS #4:CLS #5
170 REM *****
180 REM * CREATION DES TABLEAUX DE MEMORISATION *
190 REM *****
200 DIM P$(100):DIM R$(100)
210 REM *****
220 REM * INITIALISATIONS DES VARIABLES *
230 REM *****
240 LET V$="":CLS #1:CLS #2:CLS #3:CLS #4:CLS #5
250 LET SA=0:LET SB=0:LET SC=0:LET SZ=0:LET TJ=3:LET CT=1
260 REM *****
270 REM * DONNEES DE LA MUSIQUE *
280 REM *****
290 DATA 119,2,119,2
300 DATA 119,1,106,1,95,1,89,1,106,2
310 DATA 119,1,106,1,95,1,89,1,106,2
320 DATA 119,1,106,1,95,1,89,1,106,2
330 DATA 106,2,106,2
340 DATA 119,1,106,1,95,1,89,1,106,2
350 DATA 119,1,106,1,95,1,89,1,106,2
360 DATA 119,1,106,1,95,1,89,1,80,1,119,1,89,2
370 REM *****
380 REM * INTRODUCTION DU NOM DES JOUEURS *
390 REM *****
400 LOCATE #1,2,5:INPUT #1,"Premier joueur ?",A$
410 CLS #1
420 LOCATE #1,2,5: INPUT #1,"Second joueur ?",B$
430 CLS #1
440 LOCATE #1,2,5: INPUT #1,"Troisieme joueur ?",C$
450 CLS #1
460 REM *****
470 REM * AFFICHAGE DU NOM DES JOUEURS *
480 REM *****
490 PEN #3,0:LOCATE #3,2,2:PRINT #3,A$
500 PEN #4,0:LOCATE #4,2,2:PRINT #4,B$
510 PEN #5,0:LOCATE #5,2,2:PRINT #5,C$
520 REM *****
530 REM * INTRODUCTION DE LA PHRASE *
540 REM *****
550 LOCATE #1,2,2:PRINT #1,"Tapez votre phrase et"
560 LOCATE #1,2,4:PRINT #1,"validez la par ENTER."
570 LET X=2:LET Y=2
580 FOR T=1 TO 1000:NEXT T:CLS #1
590 FOR I=1 TO 100
600 LET K$= INKEY$
610 IF K$="" THEN GOTO 600
620 IF ASC(K$)=13 THEN LET I=100:GOTO 680
630 LOCATE #1,X,Y:PRINT #1,K$
640 LET P$(I)=K$
650 LET X=X+1
660 IF X>20 AND K$=" " THEN LET Y=Y+2:LET X=2
670 NEXT I
680 CLS #1
690 GOSUB 1680
700 REM *****
710 REM * INDICATION DE LA PREMIERE VOYELLE *
720 REM *****
730 LOCATE #2,2,2:INPUT #2,"VOYELLE ? ",V$
740 IF V$<>"A" AND V$<>"E" AND V$<>"I" AND V$<>"O" AND V$
<>"U" AND V$<>"Y" THEN GOTO 730
750 GOSUB 1680
760 REM *****
770 REM * DEBUT DU JEU *
780 REM *****
790 CLS #2
800 LOCATE #2,2,2:PRINT #2,"LANCEZ"
810 LOCATE #2,2,4:PRINT #2,"LA ROUE."
820 IF TJ=6 THEN LET TJ=3
830 PAPER #3,3:PAPER #4,3:PAPER #5,3
840 LOCATE #TJ,2,3:PRINT #TJ," A VOUS. "
850 PAPER #3,1:PAPER #4,1:PAPER #5,1
860 IF INKEY$="" THEN GOTO 860
870 GOSUB 1970
880 FOR T=1 TO 2000:NEXT T
890 REM *****
900 REM * TOTALISATION ET REPARTITION POINTS *
910 REM *****
920 IF TJ<>3 THEN GOTO 1100
930 LET SA=SA+SZ:IF SZ=0 THEN LET SA=0
940 LOCATE #3,2,4:PRINT #3,SA
950 IF SZ=0 THEN GOTO 1100
960 IF SA<4000 THEN GOTO 1050
970 CLS #2
980 LOCATE #2,2,2:PRINT #2,"ACHETEZ-":PRINT #2,"VOUS UNE"
:PRINT #2,"VOYELLE ?"
990 LET K$=INKEY$:IF K$="" THEN GOTO 990
1000 IF K$<>"O" THEN GOTO 1050
1010 LET SA=SA-4000
1020 LOCATE #3,2,4:PRINT #3,SA
1030 CLS #2:LOCATE #2,2,2:INPUT #2,"VOYELLE ? ",V$
1040 GOSUB 1680
1050 CLS #2:LOCATE #2,2,2:INPUT #2,"CONSONNE ? ",V$
1060 IF V$="A" OR V$="E" OR V$="I" OR V$="O" OR V$="U" OR
V$="Y" THEN GOTO 1050
1070 GOSUB 1680
1080 LET SA=SA+SZ
1090 LOCATE #3,2,4:PRINT #3,SA
1100 IF TJ<>4 THEN GOTO 1280
1110 LET SB=SB+SZ:IF SZ=0 THEN LET SB=0
1120 LOCATE #4,2,4:PRINT #4,SB
1130 IF SZ=0 THEN GOTO 1280

```



le juge en début de jeu. Le second, R\$, permettra de contrôler la réponse fournie par les participants. Nous initialiserons alors l'ensemble des variables nécessaires au bon déroulement du jeu. SA, SB et SC correspondant au score des trois joueurs ; SZ permet de déterminer le gain d'un coup, qu'il s'agisse d'un lancer de la roue ou de l'affichage de consonnes, TJ sera utilisée pour gérer le tour des participants et CT indiquera qu'une réponse fautive a été fournie. Nous trouverons ensuite, lignes 290 à 360, l'ensemble des données destinées à la musique, regroupées sous forme de chaînes DATA.

Celles-ci seront utilisées pour l'illustration sonore accompagnant la rotation de la roue. Des lignes 400 à 450, l'ordinateur demandera de lui indiquer le nom de chaque joueur puis les inscrira sur les pupitres respectifs (lignes 490 à 510). L'introduction de la phrase à dé-

couvrir sera alors demandée (lignes 550 et 560). Celle-ci s'inscrira automatiquement sur la fenêtre principale et, au fur et à mesure de sa frappe, les caractères seront mémorisés dans le tableau P\$ ; les lignes 570 à 680 se charge-

ront de cette opération.

Une fois cette frappe terminée l'ordinateur utilisera la sous-routine 1680, sous-routine que nous examinerons plus loin, pour afficher les points correspondants à l'emplacement de chaque lettre.




Photo M. Toscas

```

1140 IF SB<4000 THEN GOTO 1230
1150 CLS #2
1160 LOCATE #2,2,2:PRINT #2,"ACHETEZ-":PRINT #2,"VOUS UNE
":PRINT #2,"VOYELLE ?"
1170 LET K$=INKEY$:IF K$="" THEN GOTO 1170
1180 IF K$<>"0" THEN GOTO 1230
1190 LET SB=SB-4000
1200 LOCATE #4,2,4:PRINT #4,SB
1210 CLS #2:LOCATE #2,2,2:INPUT #2,"VOYELLE ? ",V$
1220 GOSUB 1680
1230 CLS #2:LOCATE #2,2,2:INPUT #2,"CONSONNE ? ",V$
1240 IF V$="A" OR V$="E" OR V$="I" OR V$="O" OR V$="U" OR
V$="Y" THEN GOTO 1230
1250 GOSUB 1680
1260 LET SB=SB+SZ
1270 LOCATE #4,2,4:PRINT #4,SB
1280 IF TJ<>5 THEN GOTO 1460
1290 LET SC=SC+SZ:IF SZ=0 THEN LET SC=0
1300 LOCATE #5,2,4:PRINT #5,SC
1310 IF SZ=0 THEN GOTO 1460
1320 IF SC<4000 THEN GOTO 1410
1330 CLS #2
1340 LOCATE #2,2,2:PRINT #2,"ACHETEZ-":PRINT #2,"VOUS UNE
":PRINT #2,"VOYELLE ?"
1350 LET K$=INKEY$:IF K$="" THEN GOTO 1350
1360 IF K$<>"0" THEN GOTO 1410
1370 LET SC=SC-4000
1380 LOCATE #5,2,4:PRINT #5,SC
1390 CLS #2:LOCATE #2,2,2:INPUT #2,"VOYELLE ? ",V$
1400 GOSUB 1680
1410 CLS #2:LOCATE #2,2,2:INPUT #2,"CONSONNE ? ",V$
1420 IF V$="A" OR V$="E" OR V$="I" OR V$="O" OR V$="U" OR
V$="Y" THEN GOTO 1410
1430 GOSUB 1680
1440 LET SC=SC+SZ
1450 LOCATE #5,2,4:PRINT #5,SC
1460 IF SZ=0 THEN GOTO 1630
1470 REM *****
1480 REM * DEMANDE DE LA PHRASE DEVINEE *
1490 REM *****
1500 CLS #2:LOCATE #2,2,2:PRINT #2,"UNE IDEE ?"
1510 LET K$=INKEY$:IF K$="" THEN GOTO 1510
1520 IF K$<>"0" THEN LET SZ=0:GOTO 1630
1530 REM *****
1540 REM * CONROLE DU RESULTAT, REBOUCLAGE JEU *
1550 REM *****
1560 GOSUB 1810
1570 IF CT=1 THEN SZ=0:GOTO 1630
1580 CLS #2:PRINT #2,"BRAVO !":PRINT #2,"TAPER UNE":PRINT
#2,"TOUCHE."
1590 GOSUB 1680
1600 IF INKEY$="" THEN GOTO 1600
1610 FOR N=1 TO 100:LET P$(N)="" :LET R$(N)="" :NEXT N:CLS
#1:GOTO 240
1620 LOCATE #5,2,3:PRINT #5," "
1630 IF SZ=0 THEN LOCATE #TJ,2,3:PRINT #TJ," ":LE
T TJ=TJ+1
1640 GOTO 790
1650 REM *****
1660 REM * SOUS ROUTINE D'AFFICHAGE DE LA PHRASE *
1670 REM *****
1680 LET X=2:LET Y=2:LET SZ=0
1690 FOR I=1 TO 100
1700 IF P$(I)="" THEN LET I=100:GOTO 1770
1710 IF P$(I)<>" " AND V$="" THEN LOCATE #1,X,Y:PRINT #1,
" "
1720 IF P$(I)=V$ THEN LOCATE #1,X,Y:PRINT #1,V$:LET SZ=SZ
+100
1730 IF CT=0 THEN LOCATE #1,X,Y:PRINT #1,P$(I):LET SZ=SZ+
100
1740 LET X=X+1
1750 IF X>20 AND P$(I)="" THEN LET Y=Y+2:LET X=2
1760 NEXT I
1770 RETURN
1780 REM *****
1790 REM * SOUS ROUTINE DE CONTROLE DE REPONSE *
1800 REM *****
1810 CLS #2:PRINT #2,"VOTRE":PRINT #2,"PHRASE ?":PRINT #2
1820 LET CT=0
1830 FOR I=1 TO 100
1840 LET K$=INKEY$
1850 IF K$="" THEN GOTO 1840
1860 IF ASC(K$)=13 THEN LET I=100:GOTO 1900
1870 PRINT #2,K$
1880 LET R$(I)=K$
1890 NEXT I
1900 FOR C=1 TO 100
1910 IF R$(C)<>P$(C) THEN LET CT=1
1920 NEXT C
1930 RETURN
1940 REM *****
1950 REM * SOUS ROUTINE D'ANIMATION DE LA ROUE *
1960 REM *****
1970 RESTORE 290
1980 LET G=INT(RND*6)+1:LET P=1
1990 FOR M=1 TO 36
2000 READ N:READ D
2010 SOUND 1,N,0,0,D,0,0
2020 SOUND 2,N+1,0,0,D,0,0
2030 ON P GOSUB 2150,2160,2170,2180,2190,2200
2040 LET P=P+1:IF P=7 THEN LET P=1
2050 PRINT #2
2060 PRINT #2,"*****"
2070 PRINT #2
2080 NEXT M
2090 CLS #2:LOCATE #2,1,3
2100 ON G GOSUB 2150,2160,2170,2180,2190,2200
2110 RETURN
2120 REM *****
2130 REM * SOUS ROUTINES D'AFFICHAGE DE LA ROUE *
2140 REM *****
2150 PRINT #2," PATATRAC":LET SZ=0:RETURN
2160 PRINT #2," 500 F":LET SZ=500:RETURN
2170 PRINT #2," 1 000 F":LET SZ=1000:RETURN
2180 PRINT #2," 1 500 F":LET SZ=1500:RETURN
2190 PRINT #2," 2 500 F":LET SZ=2500:RETURN
2200 PRINT #2," 5 000 F":LET SZ=5000:RETURN

```



La voyelle "offerte" pour simplifier le travail des joueurs sera demandée (lignes 730 à 750) puis la sous-routine 1680 sera de nouveau utilisée pour l'afficher aux emplacements adéquats sur la fenêtre principale.

Ainsi, totalement initialisé, le jeu peut commencer. En premier lieu le lancement de la roue sera demandé (Lignes 800 à 810) et il sera indiqué à quel participant cette demande s'adresse. En effet le pupitre du joueur en question portera la mention "à vous". Ici, la sous-routine 1970 sera utilisée. A l'issue de ce lancer viendront la totalisation et la répartition des points obtenus. La variable TJ servira d'aiguillage et affectera donc le résultat du lancer au joueur effectivement concerné. Ensuite, en fonction de son score, il lui sera demandé ou non s'il désire acheter une voyelle, puis de donner une consonne.

Si la consonne indiquée figure effectivement dans la phrase le score sera majoré grâce à l'utilisation de la sous-routine 1680. En cas contraire, cette même sous-routine indiquera à l'ordinateur qu'il est temps de passer la main au joueur suivant. Nous retrouverons trois fois de suite le même type de fonctionnement, respectivement entre les lignes 920 et 1090, 1100 et 1270 puis 1280 et 1460.

L'ordinateur demandera alors si la phrase à deviner a été découverte. Si oui, la sous-routine 1810 sera engagée et, en cas de succès, l'ordinateur affichera "BRAVO", puis la phrase complète s'affichera sur la fenêtre principale. Le jeu se terminera ainsi et la frappe d'une touche permettra d'entamer une nouvelle partie. Si à la question posée la réponse est non ou si la phrase proposée ne correspond pas à celle choisie, l'ordinateur passera au candidat suivant.

Il ne nous reste plus à présent qu'à examiner les diverses sous-routines utilisées pour le déroulement de ce jeu; elles sont au nombre de quatre.

- L'affichage de la phrase. Cette sous-routine occupe les lignes 1680 à 1770. Les variables X et Y permettent de déterminer l'emplacement de l'affichage de chaque caractère. Ensuite le tableau P\$ sera scruté à l'aide d'une boucle FOR... NEXT, et son contenu s'affi-

chera à la place déterminée par X et Y. Notons qu'en début de partie l'affichage de points sera demandé pour chaque caractère contenu par P\$ (ligne 1710). Ceci permettra de présenter, sur le tableau d'affichage principal, le nombre de lettres et de mots contenus par la phrase à découvrir. Cette sous-routine sera réutilisée dans la suite du jeu, pour mettre en place les consonnes ou les voyelles indiquées.

- Le contrôle de la réponse. Ici nous introduirons dans le tableau R\$ les caractères tapés au clavier. La variable K\$ sera donc utilisée pour détecter la frappe d'une touche et sa valeur sera transférée dans R\$. Une fois la frappe terminée, soit par l'appui de la touche ENTER, soit parce que 100 caractères auront été introduits, le programme comparera les deux tableaux R\$ et P\$. Si l'un des caractères diffère, la variable de contrôle CT prendra la valeur 1, en cas contraire elle conservera la valeur 0

- La rotation de la roue. Cette sous-routine assumera diverses fonctions. D'une part elle déterminera aléatoirement le résultat du lancer (ligne 1980) et, d'autre part, sera chargée de l'exécution de l'illustration sonore accompagnant le tirage. Pour cela, les données de la musique seront lues par la ligne 2000 et jouées par les lignes 2010 et 2020. Enfin diverses sous-routines auxiliaires seront employées pour l'affichage des cases de la roue et pour celui du résultat final. Elles occupent les lignes 2150 à 2200 du programme et sont sélectionnées soit par la ligne 2030 si la roue est en train de tourner, soit par la ligne 2100 s'il s'agit de la présentation du résultat du lancer.

La frappe de ce programme ne doit pas poser de problème particulier. Pour ceux d'entre vous qui désireraient l'adapter à d'autres machines rappelons que l'instruction WINDOW permet, sur l'Amstrad, de créer des fenêtres d'affichage sur l'écran. Par exemple WINDOW #1,5,35,2,12 permet de créer la fenêtre numéro 1 (#1) dont le bord gauche sera placé à 5 caractères du bord de l'écran, le droit à 35 caractères toujours du bord de l'écran, le haut à 2 et le bas à 12.

Nous trouverons, par la suite,

diverses commandes liées à l'affichage sur cette fenêtre. Par exemple CLS #1 n'efface que la fenêtre 1, PRINT #1,... inscrit le texte sur la fenêtre 1, il en sera de même pour LOCATE #1,X,Y ou PAPER #1, etc.

L'utilisation du jeu est également très simple. Après avoir frappé RUN l'ordinateur indiquera: "Nom du premier joueur?". Nous vous conseillons alors de passer en mode CAPS LOCK et de taper ce nom en terminant par ENTER. De même, les noms des second et troisième joueurs seront indiqués. Il faudra alors introduire la phrase à deviner. Elle sera tapée sans se préoccuper de sa mise en place sur l'écran; en effet l'ordinateur se charge de la paginer automatiquement.

Si le nombre de caractères qu'elle contient est inférieur à 100, il suffira de la valider en fin de frappe par ENTER. En cas contraire, l'ordinateur la mémorisera automatiquement une fois la centième lettre frappée. Cette opération terminée l'Amstrad demandera de lui indiquer la voyelle offerte en début de jeu. Celle-ci sera donc indiquée puis validée par ENTER. Le jeu pourra alors commencer.

Pour cela, il faut lancer la roue en appuyant sur une touche quelconque du clavier. A l'issue du tirage, le score du joueur concerné sera crédité du résultat obtenu, sauf en cas de "Patatrac", et l'ordinateur demandera de lui fournir une consonne. Enfin, la possibilité de donner la phrase sera offerte. Si la réponse à cette offre est "NON", la main passera au joueur suivant, en cas contraire la réponse sera introduite au clavier puis vérifiée par l'ordinateur. Si elle est fautive la main passera, si elle est exacte le joueur sera déclaré gagnant. Un dernier cas concerne l'achat de voyelles.

En effet, si le score de l'un des participants, après lancer de la roue, est supérieur à 4000, il lui sera demandé s'il désire acheter une voyelle. La réponse pourra être OUI ou NON. Si celle-ci est OUI, son crédit sera diminué de 4000 points et l'ordinateur demandera de lui indiquer la voyelle choisie. En cas contraire, cette possibilité sera ignorée et le jeu se poursuivra.

**Henri-Pierre Penel**